STATISTICAL HORIZONS

Multilevel Structural Equation Modeling

Kristopher Preacher, Ph.D.

Upcoming Seminar:

July 29-August 2, 2019, Philadelphia, Pennsylvania

STATISTICAL HORIZONS

A simple example in Mplus

The first MLM example uses the High School and Beyond (HSAB)* data.

N = 7,185 students, J = 160 schools

school	School ID
minority	1 = minority, 0 = other
female	1 = female, 0 = male
ses	parent socioeconomic status
mathach	mathematics achievement
size	school enrollment
sector	1 = Catholic, 0 = public
meanses	school mean SES

*http://www.ats.ucla.edu/stat/hlm/seminars/hlm6/mlm_hlm6_seminar.htm

67

STATISTICAL HORIZONS

Random effects ANOVA ("random intercepts only" model)

$$mathach_{ij} = \gamma_{00} + u_{0j} + e_{ij} \qquad \begin{aligned} u_{0j} &\sim N(0, \tau_{00}) \\ e_{ij} &\sim N(0, \sigma^2) \end{aligned}$$

A random effects ANOVA model simply splits $mathach_{ij}$ into "within" and "between" components.

Because these components are uncorrelated, the variance of the outcome is also split into two components that add to yield the total variance of $mathach_{ij}$.

$$\text{var}(mathach_{ij}) = \tau_{00} + \sigma^2$$

68

Random effects ANOVA ("random intercepts only" model)

The principle reasons to carry out a random effects ANOVA are to estimate the variance components, estimate the ICC, and to serve as a baseline "null" model for adding predictors.

Random effects ANOVA ("random intercepts only" model)

```
TITLE: hsab random effects anova;
DATA: FILE IS hsab.dat;
VARIABLE:
NAMES ARE school minority female
ses mathach size sector meanses;
USEVARIABLES ARE mathach;
CLUSTER IS school;
ANALYSIS: TYPE IS TWOLEVEL;
MODEL:
```

Random effects ANOVA (“random intercepts only” model)

```
TITLE: hsab random effects anova;
DATA: FILE IS hsab.dat;
VARIABLE:
NAMES ARE school minority female
ses mathach size sector meanses;
USEVARIABLES ARE mathach;
CLUSTER IS school;
ANALYSIS: TYPE IS TWOLEVEL;
MODEL:

%WITHIN%
mathach;

%BETWEEN%
mathach; [mathach];
```



Random effects ANOVA (“random intercepts only” model)

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Variances				
MATHACH	39.148	0.835	46.876	0.000
Between Level				
Means				
MATHACH	12.637	0.244	51.824	0.000
Variances				
MATHACH	8.562	1.057	8.101	0.000

Random intercept; level-1 and level-2 predictors

$$mathach_{ij} = \beta_{0j} + \beta_{1j} ses_{ij} + e_{ij}$$

$$\beta_{0j} = \gamma_{00} + \gamma_{01} sector_j + u_{0j}$$

$$\beta_{1j} = \gamma_{10}$$

$$u_{0j} \sim N(0, \tau_{00})$$

$$e_{ij} \sim N(0, \sigma^2)$$

This model includes SES as a level-1 predictor and school sector (public vs. Catholic) as a level-2 predictor.

Intercepts are random. τ_{00} now represents residual variance at level-2.

ses_{ij} can potentially explain "within" and "between" variance in $mathach_{ij}$, but $sector_j$ can potentially explain only "between" variance.

Random intercept; level-1 and level-2 predictors

```
TITLE: hsab random intercept, L1 and L2 predictors;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority female
ses mathach size sector meanses;
USEVARIABLES ARE ses mathach sector;
CLUSTER IS school;
WITHIN IS ses; BETWEEN IS sector;
ANALYSIS: TYPE IS TWOLEVEL;
MODEL:

%WITHIN%
mathach; mathach ON ses;

%BETWEEN%
[mathach]; mathach;
mathach ON sector;
```



STATISTICAL HORIZONS

Random intercept; level-1 and level-2 predictors

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
MATHACH ON				
SES	2.376	0.128	18.589	0.000
Residual Variances				
MATHACH	37.032	0.717	51.665	0.000
Between Level				
MATHACH ON				
SECTOR	2.101	0.347	6.046	0.000
Intercepts				
MATHACH	11.719	0.226	51.860	0.000
Residual Variances				
MATHACH	3.627	0.588	6.165	0.000

hsab.randint.fixslp.inp

STATISTICAL HORIZONS

Random intercept *and slope*; level-1 and level-2 predictors

$$mathach_{ij} = \beta_{0j} + \beta_{1j} ses_{ij} + e_{ij}$$
$$\beta_{0j} = \gamma_{00} + \gamma_{01} sector_j + u_{0j}$$
$$\beta_{1j} = \gamma_{10} + u_{1j}$$

$$\begin{bmatrix} u_{0j} \\ u_{1j} \end{bmatrix} \sim MVN\left[\begin{pmatrix} 0 \\ 0 \end{pmatrix}, \begin{pmatrix} \tau_{00} & \\ \tau_{10} & \tau_{11} \end{pmatrix}\right]$$
$$e_{ij} \sim N\left(0, \sigma^2\right)$$

This model includes SES as a level-1 predictor and school sector (public vs. Catholic) as a level-2 predictor.

Intercepts and slopes are now random.

Random intercept *and slope*; level-1 and level-2 predictors

```
TITLE: hsab random intercept and slope, L1 and L2 predictors;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority female
ses mathach size sector meanses;
USEVARIABLES ARE ses mathach sector;
CLUSTER IS school;
WITHIN IS ses; BETWEEN IS sector;
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL:

%WITHIN%
mathach; s1 | mathach ON ses;

%BETWEEN%
[mathach s1];
mathach s1; mathach WITH s1;
mathach ON sector;
```



Random intercept *and slope*; level-1 and level-2 predictors

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Residual Variances				
MATHACH	36.783	0.723	50.902	0.000
Between Level				
MATHACH ON				
SECTOR	2.530	0.423	5.979	0.000
MATHACH WITH				
S1	0.699	0.356	1.964	0.050
Means				
S1	2.385	0.128	18.707	0.000
Intercepts				
MATHACH	11.468	0.282	40.610	0.000
Variances				
S1	0.421	0.234	1.798	0.072
Residual Variances				
MATHACH	3.864	0.644	5.996	0.000

$$\mathbf{T} = \begin{bmatrix} \tau_{00} & \\ \tau_{10} & \tau_{11} \end{bmatrix}$$

$$= \begin{bmatrix} 3.864 & \\ .699 & .421 \end{bmatrix}$$

STATISTICAL HORIZONS

Power Analysis for MLM

STATISTICAL HORIZONS

Power: The probability of correctly rejecting a false null hypothesis, given that a particular alternative hypothesis is true. Defined only when the null hypothesis is false.

We want power to be high. Power increases with...

- higher α (Type I error rate)
- larger sample size
- smaller error variance
- larger effect size

Of these, sample size is most directly under our control. Researchers usually want to know either:

1. how much power they can obtain with a given *N*.
2. how large an *N* is necessary to achieve a given level of power.

There is a large literature on power and sample size in MLM.

The best advice: **If you can get more data, do so.** MLM is a "large-sample" technique, meaning that assumptions are met to a greater degree with more confidence as *N* increases. So, the larger the sample, the better.

How large is "large"?

Singer & Willett (2003): "10 is certainly small and 100,000 is certainly large."

Mplus' Monte Carlo facility has many uses. One is to conduct power analysis for individual parameters.

The basic idea:

- Specify a population model, with values given to all parameters.
- Decide on level-1 and level-2 sample sizes (J and n_j).
- Generate a large number of samples.
- Run the model on each of them (estimate the parameters).
- Find the proportion of these runs in which the estimate is significant.
- This proportion is empirical *power*.

Power analysis for individual parameters

Say we intend to fit the random intercept / random slope model from earlier.

We want to know how many clusters (and of what size) to aim for to have sufficient power to detect the effect of w_1 on β_{0j} when the rest of the model has been fully specified with reasonable parameter values.



Power analysis for individual parameters

```
TITLE: hsab random intercept and slope (power analysis);
MONTECARLO: NAMES ARE ses mathach sector;
NOBSERVATIONS = 1000;
CSIZES = 50(20); NREPS = 500;
SEED = 3729; NCSIZES = 1;
WITHIN IS ses; BETWEEN IS sector;
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL POPULATION:
%WITHIN%
mathach*37; ses*20;
s1 | mathach ON ses;
%BETWEEN%
[mathach*11.5 s1*.4];
mathach*3.9 s1*.5; mathach WITH s1*.7;
mathach ON sector*1.5; sector*.25;

MODEL:
%WITHIN%
mathach*37; ses*20;
s1 | mathach ON ses;
%BETWEEN%
[mathach*11.5 s1*.4];
mathach*3.9 s1*.5; mathach WITH s1*.7;
mathach ON sector*1.5; sector*.25;
```



STATISTICAL HORIZONS

Power analysis for individual parameters

```
                              ESTIMATES              S. E.     M. S. E.  95%  % Sig
                 Population   Average   Std. Dev.   Average              Cover Coeff
Within Level

 Means
  SES                 0.000    -0.0023     0.1396     0.1401     0.0195 0.934 0.066

 Variances
  SES                20.000    19.9692     0.8801     0.8787     0.7740 0.930 1.000

 Residual Variances
  MATHACH            37.000    36.9516     1.8568     1.7077     3.4431 0.934 1.000

Between Level

 MATHACH    ON
  SECTOR              1.500     1.4762     0.6717     0.6148     0.4508 0.928 0.656

 MATHACH  WITH
  S1                  0.700     0.6768     0.2808     0.2704     0.0792 0.932 0.750
...
```

hsab.randint.randslp.power.inp 85

STATISTICAL HORIZONS

Power analysis for individual parameters

To triangulate on appropriate level-1 and level-2 sample sizes, it may be useful to create a chart like this one:

J	n_J	power
50	10	.52
75	10	.68
100	10	.79
50	20	.66
75	20	.80
100	20	.90

These two have the same total sample size, but different empirical power estimates.

hsab.randint.randslp.power.inp 86

STATISTICAL HORIZONS

Multilevel EFA

STATISTICAL HORIZONS

6. Multilevel exploratory factor analysis (MEFA)

Returning to the 8-item math/science MCFA… what if we had little a priori idea which items would load on which factor? Or even how many factors existed at each level?

We could use *multilevel exploratory factor analysis* (MEFA), the multilevel analog of EFA.

We may want to entertain the possibility that either 1 or 2 factors account for the "between" covariances, and either 1 or 2 factors account for the "within" covariances.

STATISTICAL HORIZONS

6. Multilevel exploratory factor analysis (MEFA)

```
TITLE: timss mefa;
DATA: FILE IS G8_MERGED07_SMALL.csv;

DEFINE: !rescale, reverse-score, and name variables

!plausible values
y1=BSMALG01/100; !algebra
y2=BSMDAT01/100; !data/chance
y3=BSMNUM01/100; !number
y4=BSMGEO01/100; !geometry
y5=BSSCHE01/100; !chemistry
y6=BSSEAR01/100; !earth science
y7=BSSBIO01/100; !biology
y8=BSSPHY01/100; !physics

VARIABLE: NAMES ARE IDSCHOOL BSMSCPT BSMMAT01 BSSSCI01 BSMALG01 BSMDAT01 BSMNUM01
BSMGEO01 BSSCHE01 BSSEAR01 BSSBIO01 BSSPHY01 BSMREA01 BC4MSOEM BC4SSOES USBCGQ06
BSDGEDUP BSDMSVM BSDSSVS BSDMSCM;
```

STATISTICAL HORIZONS

6. Multilevel exploratory factor analysis (MEFA)

```
USEVARIABLES ARE y1-y8;
CLUSTER IS IDSCHOOL;

MISSING ARE BSMALG01 BSMDAT01 BSMNUM01 BSMGEO01
            BSSCHE01 BSSEAR01 BSSBIO01 BSSPHY01 (998 999);

ANALYSIS: TYPE IS TWOLEVEL EFA 1 2 1 2;
```

Do EFA

Fit from 1 to 2 factors in the "within" model

Fit from 1 to 2 factors in the "between" model

Note: No MODEL statement necessary here!

STATISTICAL HORIZONS

6. Multilevel exploratory factor analysis (MEFA)

With these commands, Mplus will fit the following 4 models:

1 "within" factor and 1 "between" factor
2 "within" factors and 1 "between" factor
1 "within" factor and 2 "between" factors
2 "within" factors and 2 "between" factors

Mplus provides several rotation methods. The default is oblique GEOMIN, which has shown good performance in comparison studies. Oblique GEOMIN will be used for any 2-factor solutions.

6. Multilevel exploratory factor analysis (MEFA)

1 “within” factor and 1 “between” factor

```
Y1 within      0.621*
Y2    "        0.589*
Y3    "        0.635*
Y4    "        0.642*
Y5    "        0.877*
Y6    "        0.922*
Y7    "        0.901*
Y8    "        0.935*

Y1 between     0.988*
Y2    "        1.000*
Y3    "        0.993*
Y4    "        0.996*
Y5    "        0.992*
Y6    "        0.993*
Y7    "        0.994*
Y8    "        0.991*
```

RMSEA = .215

CFI = .773
TLI = .682

timss.mefa.inp

6. Multilevel exploratory factor analysis (MEFA)

2 “within” factors and 1 “between” factor

```
Y1 within      0.856*        0.005
Y2    "        0.833*       -0.011
Y3    "        0.846*        0.028*
Y4    "        0.906*       -0.006
Y5    "        0.037*        0.851*
Y6    "       -0.029*        0.951*
Y7    "        0.002         0.904*
Y8    "        0.001         0.940*    r = .646*

Y1 between     0.990*
Y2    "        0.995*
Y3    "        0.995*
Y4    "        0.996*
Y5    "        0.993*
Y6    "        0.997*
Y7    "        0.997*
Y8    "        0.995*
```

RMSEA = .048

CFI = .991
TLI = .984

timss.mefa.inp

STATISTICAL HORIZONS

6. Multilevel exploratory factor analysis (MEFA)

1 “within” factor and 2 “between” factors

Y1 within	0.628*			
Y2 "	0.595*			
Y3 "	0.644*			
Y4 "	0.651*			
Y5 "	0.874*			
Y6 "	0.919*			
Y7 "	0.898*			
Y8 "	0.932*			
Y1 between	1.081*	-0.085		
Y2 "	0.800*	0.208*		
Y3 "	1.000*	0.000		
Y4 "	0.956*	0.046		
Y5 "	0.107*	0.891*		
Y6 "	-0.009	1.006*		
Y7 "	-0.001	0.999*		
Y8 "	0.004	0.992*	r = .956*	

RMSEA = .233

CFI = .780
TLI = .626

Artifactual second factor arising from misspecified level-1 model?

timss.mefa.inp

STATISTICAL HORIZONS

6. Multilevel exploratory factor analysis (MEFA)

2 “within” factors and 2 “between” factors

Y1 within	0.852*	0.008	
Y2 "	0.839*	-0.018	
Y3 "	0.844*	0.030*	
Y4 "	0.906*	-0.006	
Y5 "	0.034*	0.853*	
Y6 "	-0.029*	0.950*	
Y7 "	0.003	0.902*	
Y8 "	0.000	0.940*	r = .654*
Y1 between	0.941*	0.223*	
Y2 "	0.981*	0.086*	
Y3 "	0.956*	0.180*	
Y4 "	0.962*	0.161*	
Y5 "	0.994*	-0.008*	
Y6 "	1.004*	-0.061*	
Y7 "	1.004*	-0.062*	
Y8 "	1.000*	-0.040*	r = .135*

RMSEA = .050

CFI = .992
TLI = .983

Artifactual second factor disappears when level-1 model is properly specified!

timss.mefa.inp